
Lecture Outline
RDF stores

• IntroducƟon
• Linked Data

SPARQL query language
• Graph paƩerns
• Filter constraints
• SoluƟon modifiers
• AggregaƟon
• Query forms

RDF Stores
Data model

• RDF triples
Components: subject, predicate, and object
Each triple represents a statement about a real-world enƟty

• Triples can be viewed as graphs
VerƟces for subjects and objects
Edges directly correspond to individual statements

Query language
• SPARQL: SPARQL Protocol and RDF Query Language

RepresentaƟves
• Apache Jena, rdf4j (Sesame), Algebraix
• MulƟ-model: MarkLogic, OpenLink Virtuoso

Linked Data
Linked Data

• Method of publishing structured and interlinked data in a
way that allows for an automated processing by programs
rather than browsing by human readers

Principles of Linked Open Data
• IdenƟfy resources using URIs or even beƩer using URLs
• Publish data about resources in standard formats via HTTP
• Mutually interlink resources to form Web of Data
• Release the data under an open licence

Linked Open Data Cloud
May 2007

• 12 datasets

October 2007
• 25 datasets, 2 billion triples, 2 million links

Source: hƩp://lod-cloud.net/

Linked Open Data Cloud
September 2011

• 295 datasets, 31 billion triples, 504 million links

Source: hƩp://lod-cloud.net/

Linked Open Data Cloud
March 2019

• 1239 datasets, 16 thousand dataset links

Source: hƩp://lod-cloud.net/

MIE-PDB.16: Advanced Database Systems | Lecture 6: RDF Stores: SPARQL | 27. 10. 2020 7



SPARQL Query Language



SPARQL
SPARQL Query Language

• Query language for RDF data
Graph paƩerns, opƟonal graph paƩerns, subqueries, negaƟon,
aggregaƟon, value constructors, …

• Versions: 1.0 (2008), 1.1 (2013)
• W3C recommendaƟons

hƩps://www.w3.org/TR/sparql11-query/
Altogether 11 parts: query language, update facility, federated
queries, protocol, result formats, …

Sample Data
Graph ofmovies <http://db.cz/movies>

@prefix i: <http://db.cz/terms#> .
@prefix m: <http://db.cz/movies/> .
@prefix a: <http://db.cz/actors/> .
m:vratnelahve

rdf:type i:Movie ; i:title "Vratné lahve" ;
i:year 2006 ;
i:actor a:sverak , a:machacek .

m:samotari
rdf:type i:Movie ; i:title "Samotáři" ;
i:year 2000 ;
i:actor a:schneiderova , a:trojan , a:machacek .

m:medvidek
rdf:type i:Movie ; i:title "Medvídek" ;
i:year 2007 ;
i:actor a:machacek , a:trojan ;
i:director "Jan Hřebejk" .

m:zelary
rdf:type i:Movie .

Sample Data
Graph of actors <http://db.cz/actors>

@prefix i: <http://db.cz/terms#> .
@prefix a: <http://db.cz/actors/> .
a:trojan

rdf:type i:Actor ;
i:firstname "Ivan" ; i:lastname "Trojan" ;
i:year 1964 .

a:machacek
rdf:type i:Actor ;
i:firstname "Jiří" ; i:lastname "Macháček" ;
i:year 1966 .

a:schneiderova
rdf:type i:Actor ;
i:firstname "Jitka" ; i:lastname "Schneiderová" ;
i:year 1973 .

a:sverak
rdf:type i:Actor ;
i:firstname "Zdeněk" ; i:lastname "Svěrák" ;
i:year 1936 .

Sample Query
Find all movies, return their Ɵtles and years they were filmed
PREFIX i: <http://db.cz/terms#>
SELECT ?t ?y
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie ;

i:title ?t ;
i:year ?y .

}
ORDER BY ?y

?t ?y
Samotáři 2000

Vratné lahve 2006
Medvídek 2007

Sample Query

Sample Query

Graph PaƩern Matching
Graph paƩerns

• Basic graph paƩern
Based on ordinary triples with variables

– ?variable or $variable
• More complicated graph paƩerns

E.g. group, opƟonal, minus, …
Graph paƩern matching

• Our goal is to find all subgraphs of the data graph
that are matched by the query graph paƩern

I.e. subgraphs of the data graph that are idenƟcal to the query
graph paƩern with variables subsƟtuted by parƟcular terms

• Onematching subgraph = one soluƟon = one row of a table

Graph PaƩern Matching
Query result = soluƟon sequence = ordered mulƟset of soluƟons

?t ?y
Samotáři 2000

Vratné lahve 2006
Medvídek 2007

{
{ (?t, "Samotáři"), (?y, "2000") },
{ (?t, "Vratné lahve"), (?y, "2006") },
{ (?t, "Medvídek"), (?y, "2007") }

}

SoluƟon = set of variable bindings
?t ?y

Samotáři 2000
{ (?t, "Samotáři"), (?y, "2000") }

Variable binding = pair of a variable name and a value it is assigned
?t

Samotáři
(?t, "Samotáři")

Graph PaƩern Matching
CompaƟbility of soluƟons

• Two soluƟons are mutually compaƟble if and only if all the
variables they share are pairwise bound to idenƟcal values

Examples
• CompaƟble soluƟons

{ (?m, m:samotari), (?t, "Samotáři") }
{ (?m, m:samotari), (?y, "2000") }

• IncompaƟble soluƟons
{ (?m, m:samotari), (?t, "Samotáři") }
{ (?m, m:medvidek), (?y, "2007") }

Select Queries
SELECT queries

prologue declarationsprologue declarations

SELECT clauseSELECT clause

FROM clauseFROM clause

WHERE clauseWHERE clause

solution modifierssolution modifiers

• Prologue declaraƟons – PREFIX, BASE
• Main clauses

SELECT – variables to be projected
FROM – data graphs to be queried
WHERE – graph paƩerns to be matched

• SoluƟon modifiers – ORDER BY, …

Prologue DeclaraƟons
Prologue declaraƟons

• Allow to simplify IRI references by declaring base IRIs

BASEBASE IRI referenceIRI reference

PREFIXPREFIX prefix nameprefix name :: IRI referenceIRI reference

BASE clause
• At most one base IRI can be defined
• All relaƟve IRI references are then related to this base IRI

PREFIX clause
• Several base IRIs are defined, each is associated with a name

These names must be disƟnct, one of them can be empty
• All prefixed names are then related to the respecƟve base IRIs

Prologue DeclaraƟons
Examples

• When BASE <http://db.cz/> is defined,
then a relaƟve IRI reference terms#Movie
is interpreted as http://db.cz/terms#Movie

• When PREFIX i: <http://db.cz/> is defined,
then a prefixed name i:terms#Movie
is interpreted as http://db.cz/terms#Movie

Where Clause
WHERE clause

• Describes one group graph paƩern
WHEREWHERE group graph patterngroup graph pattern

Types of graph paƩerns
• Basic – triple paƩerns to be matched
• Group – set of graph paƩerns to be matched
• OpƟonal – graph paƩern to be matched only if possible
• AlternaƟve – two or more alternaƟve graph paƩerns
• …

Graph paƩerns can be inducƟvely combined into complex ones

Basic Graph PaƩern
Basic graph paƩern (block of triples)

One or more triple paƩerns to be all matched

• Ordinary triples separated by .
E.g. s p1 o1 . s p1 o2 . s p2 o3 .

• May contain variables
E.g. ?var or $var

• Turtle abbreviaƟons also permiƩed
Object lists using , or predicate-object lists using ;

– E.g. s p1 o1 , o2 ; p2 o3 .
Blank nodes using []

– Act as non-selectable variables
– I.e. do not enforce to be matched only by blank nodes in data!

Basic Graph PaƩern
InterpretaƟon

• All the involved triples must be matched
I.e. we combine them as if they were in conjuncƟon
More precisely…

– Each triple paƩern is evaluated to its soluƟon sequence
– All combinaƟons of compaƟble soluƟons are then found

• Note that all the variables need to be bound
I.e. if any of the involved variables cannot be bound at all,
then the enƟre basic graph paƩern cannot be matched!

Basic Graph PaƩern: Example
Titles and years of all movies
PREFIX i: <http://db.cz/terms#>
SELECT ?t ?y
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie . # triple 1
?m i:title ?t . # triple 2
?m i:year ?y . # triple 3

}

?t ?y
Vratné lahve 2006
Samotáři 2000
Medvídek 2007

Basic Graph PaƩern: Example

Jt1K =
?m

m:vratnelahve
m:samotari
m:medvidek
m:zelary

Jt2K = ?m ?t
m:vratnelahve Vratné lahve
m:samotari Samotáři
m:medvidek Medvídek

Jt3K = ?m ?y
m:vratnelahve 2006
m:samotari 2000
m:medvidek 2007

?t ?y
Vratné lahve 2006
Samotáři 2000
Medvídek 2007

Equality of Terms
IRIs
Literals

• Plain values must be idenƟcal
• And when types / language tags are specified,

these must be idenƟcal as well
E.g.: "Medvídek"@cs != "Medvídek"

Blank nodes
• Blank nodes in query paƩerns act as non-selectable variables
• Labels of blank nodes in

data graphs / query graph paƩerns / query results
may not refer to the same nodes despite being the same

I.e. the scope of validity is always local only

MIE-PDB.16: Advanced Database Systems | Lecture 6: RDF Stores: SPARQL | 27. 10. 2020 26



Group Graph PaƩern
Group graph paƩern

Set of graph paƩerns to be all matched

{{ nested SELECT querynested SELECT query

triplestriples group graph patterngroup graph pattern

OPTIONAL graph patternOPTIONAL graph pattern

UNION graph patternUNION graph pattern

MINUS graph patternMINUS graph pattern

GRAPH graph patternGRAPH graph pattern

BINDBIND

FILTERFILTER

triplestriples

}}

Group Graph PaƩern
Two modes

• Nested SELECT query
Only with SELECT and WHERE clauses and soluƟon modifiers
i.e. without FROM clause

• Set of graph paƩerns interleaved by triple blocks
InterpretaƟon

• All the involved graph paƩerns must be matched
I.e. we combine them as if they were in conjuncƟon

Notes
• Empty group paƩerns {} are also allowed

OpƟonal Graph PaƩern
OPTIONAL graph paƩern

One group graph paƩern is tried to be matched

OPTIONALOPTIONAL group graph patterngroup graph pattern

InterpretaƟon
• When the opƟonal part does not match,
it creates no bindings but does not eliminate the soluƟon

OpƟonal Graph PaƩern: Example
Movies together with their directors when possible
PREFIX i: <http://db.cz/terms#>
SELECT ?t ?y ?d
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie ;

i:title ?t ;
i:year ?y .

OPTIONAL { ?m i:director ?d . }
}

?t ?y ?d
Vratné lahve 2006
Samotáři 2000
Medvídek 2007 Jan Hřebejk

AlternaƟve Graph PaƩern
UNION graph paƩern

Two or more group graph paƩerns are to be matched

group graph patterngroup graph pattern UNIONUNION group graph patterngroup graph pattern

InterpretaƟon
• Standard set union of the involved query results

Minus Graph PaƩern
MINUS graph paƩern

One group graph paƩern removing compaƟble soluƟons

MINUSMINUS group graph patterngroup graph pattern

InterpretaƟon
• SoluƟons of the first paƩern are preserved if and only if they
are not compatible with any soluƟon of the second paƩern

I.e. minus graph paƩern does not correspond to the standard
set minus operaƟon!

Minus Graph PaƩern: Example
Titles of movies that have no director
PREFIX i: <http://db.cz/terms#>
SELECT ?t
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie ;

i:title ?t . # pattern 1
MINUS { ?m rdf:type i:Movie ; i:director ?d . } # pattern 2

}

?t
Vratné lahve
Samotáři

Minus Graph PaƩern: Example

Jp1K = ?m ?t
m:vratnelahve Vratné lahve
m:samotari Samotáři
m:medvidek Medvídek

Jp2K = ?m ?d
m:medvidek Jan Hřebejk

?t
Vratné lahve
Samotáři

From Clause
FROM clause

• Defines data graphs to be queried
FROMFROM

NAMEDNAMED

IRI referenceIRI reference

prefixed nameprefixed name

Dataset = collecƟon of graphs to be queried
• One default graph

Merge of all the declared graphs from unnamed FROM clauses
Empty when no unnamed FROM clause is provided

• Zero or more named graphs
AcƟve graph = used for the evaluaƟon of graph paƩerns

• The default graph unless changed using GRAPH graph paƩern

From Clause: Example
Names of actors who played inMedvídekmovie
PREFIX i: <http://db.cz/terms#>
PREFIX m: <http://db.cz/movies/>
SELECT ?f ?l
FROM <http://db.cz/movies>
FROM <http://db.cz/actors>
WHERE

{
m:medvidek i:actor ?a .
?a i:firstname ?f ; i:lastname ?l .

}

?f ?l
Jiří Macháček
Ivan Trojan

Graph Graph PaƩern
GRAPH graph paƩern

PaƩern evaluated with respect to a parƟcular named graph

GRAPHGRAPH variablevariable

IRI referenceIRI reference

prefixed nameprefixed name

group graph patterngroup graph pattern

• Changes the acƟve graph for a given group graph paƩern
GRAPH <http://db.cz/actors> { … }

• We can also consider all the named graphs
GRAPH ?g { … }

Graph Graph PaƩern: Example
Names of actors who played inMedvídekmovie
PREFIX i: <http://db.cz/terms#>
PREFIX m: <http://db.cz/movies/>
SELECT ?f ?l
FROM <http://db.cz/movies>
FROM NAMED <http://db.cz/actors>
WHERE

{
m:medvidek i:actor ?a .
GRAPH <http://db.cz/actors> {

?a i:firstname ?f ; i:lastname ?l .
}

}

?f ?l
Jiří Macháček
Ivan Trojan

Variable Assignments
BIND graph paƩern

Explicitly assigns a value to a given variable

BINDBIND (( expressionexpression ASAS variablevariable ))

• This variable must not yet be bound!

Filter Constraints
FILTER constraints

Impose constraints on variables and their values

FILTERFILTER built-in callbuilt-in call

(( expressionexpression ))

• Only soluƟons saƟsfying the given condiƟon are preserved
• Does not create any new variable bindings!
• Always applied on the enƟre group graph pattern

i.e. evaluated at the very end

Filter Constraints: Example
Movies filmed in 2005 or later where Ivan Trojan played
PREFIX i: <http://db.cz/terms#>
PREFIX a: <http://db.cz/actors/>
SELECT ?t ?y
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie ;

i:title ?t ;
i:year ?y .

FILTER (
(?y >= 2005) &&
EXISTS { ?m i:actor a:trojan . }

)
}

?t ?y
Medvídek 2007

Filter Constraints
RelaƟonal expressions

• Comparisons
=, !=, <, <=, =>, >
Unbound variable < blank node < IRI < literal

• Set membership tests
IN and NOT IN

NOTNOT

ININ ((

expressionexpression

,,

))

Numeric expressions
• Unary / binary arithmeƟc operators +, -, *, /

Filter Constraints
Primary expressions

• Literals – numeric, boolean, RDF triples
• Variables
• Built-in calls
• Parentheses

Boolean expressions
• Logical connecƟves

ConjuncƟon &&, disjuncƟon ||, negaƟon !
• 3-value logic because of unbound variables (NULL values)

true, false, error

Filter Constraints
Built-in calls

• Term accessors
STR – lexical form of an IRI or literal
LANG – language tag of a literal
DATATYPE – data type of a literal

• Variable tests
BOUND – true when a variable is bound to a value
isIRI, isBLANK, isLITERAL

Filter Constraints
Built-in calls

• Existence tests
EXISTS

– True when a provided group graph paƩern
is evaluated to at least one soluƟon

NOT EXISTS

NOTNOT

EXISTSEXISTS group graph patterngroup graph pattern

Select Clause
SELECT clause

• Enumerates variables to be included in the query result
SELECTSELECT

DISTINCTDISTINCT

REDUCEDREDUCED

variablevariable

expressionexpression ASAS variablevariable

**

• Asterisk * selects all the variables
SoluƟon modifiers

• DISTINCT – duplicate soluƟons are removed
• REDUCED – some duplicate soluƟons may be removed

(implementaƟon-dependent behavior)

SoluƟon Modifiers
SoluƟon modifiers – modify the enƟre soluƟon sequence

• AggregaƟon
GROUP BY and HAVING

• Ordering
ORDER BY
LIMIT and OFFSET

GROUP BY clauseGROUP BY clause HAVING clauseHAVING clause

ORDER BY clauseORDER BY clause LIMIT clauseLIMIT clause

OFFSET clauseOFFSET clause

OFFSET clauseOFFSET clause

LIMIT clauseLIMIT clause

SoluƟon Modifiers
ORDER BY clause

• Defines the order of soluƟons within the query result

ORDER BYORDER BY variablevariable

built-in callbuilt-in call

(( expressionexpression ))

ASCASC

DESCDESC

(( expressionexpression ))

• ASC(…) = ascending (default)
• DESC(…) = descending

SoluƟon Modifiers
OFFSET clause

• Skips a certain number of soluƟons in the query result

OFFSETOFFSET integerinteger

LIMIT clause
• Limits the number of soluƟons in the query result

LIMITLIMIT integerinteger

SoluƟon Modifiers: Example
PREFIX i: <http://db.cz/terms#>
SELECT ?t ?y
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie ;

i:title ?t ;
i:year ?y .

}
ORDER BY DESC(?y) ASC(?t)
OFFSET 1
LIMIT 5

?t ?y
Vratné lahve 2006
Samotáři 2000

AggregaƟon
GROUP BY + HAVING clauses

• Standard aggregaƟon over a soluƟon sequence

GROUP BYGROUP BY variablevariable

built-in callbuilt-in call

(( expressionexpression

ASAS variablevariable

))

HAVINGHAVING built-in callbuilt-in call

(( expressionexpression ))

AggregaƟon: Example
Numbers of actors in movies with at most 2 actors
PREFIX i: <http://db.cz/terms#>
SELECT ?t (COUNT(?a) AS ?c)
FROM <http://db.cz/movies>
WHERE

{
?m rdf:type i:Movie ;

i:title ?t ;
i:actor ?a .

}
GROUP BY ?m ?t
HAVING (?c <= 2)
ORDER BY ?c ?t

?t ?c
Medvídek 2

Vratné lahve 2

AggregaƟon
Aggregate funcƟons

COUNTCOUNT ((

DISTINCTDISTINCT

expressionexpression

**

))

SUMSUM

MINMIN

MAXMAX

AVGAVG

((

DISTINCTDISTINCT

expressionexpression ))

GROUP_CONCATGROUP_CONCAT ((

DISTINCTDISTINCT

expressionexpression

;; SEPARATORSEPARATOR == stringstring

))

Query Forms
Query forms

• SELECT
Finds soluƟons matching a provided graph paƩern

• ASK
Checks whether at least one soluƟon exists

• DESCRIBE
Retrieves a graph with data about selected resources

• CONSTRUCT
Creates a new graph according to a provided paƩern

Select Query Form
SELECT query form

Finds soluƟons matching a provided graph paƩern

prologue declarationsprologue declarations

SELECT clauseSELECT clause

FROM clauseFROM clause

WHERE clauseWHERE clause

solution modifierssolution modifiers

Result
• SoluƟon sequence = ordered mulƟset of soluƟons

Construct Query Form
CONSTRUCT query form

Creates a new graph according to a provided paƩern

prologue declarationsprologue declarations

CONSTRUCTCONSTRUCT {{ triples constructiontriples construction }}

FROM clauseFROM clause

WHERE clauseWHERE clause

solution modifierssolution modifiers

Result
• RDF graph constructed according to a group graph paƩern

Unbound or invalid triples are not involved

Construct Query Form: Example
PREFIX i: <http://db.cz/terms#>
CONSTRUCT

{
?a i:name concat(?f, " ", ?l) .

}
FROM <http://db.cz/actors>
WHERE

{
?a rdf:type i:Actor ;

i:firstname ?f ;
i:lastname ?l .

}

<http://db.cz/actors/trojan> i:name "Ivan Trojan" .
<http://db.cz/actors/machacek> i:name "Jiří Macháček" .
<http://db.cz/actors/schneiderova> i:name "Jitka Schneiderová" .
<http://db.cz/actors/sverak> i:name "Zdeněk Svěrák" .

Lecture Conclusion
SPARQL

• Query forms
SELECT, ASK, DESCRIBE, CONSTRUCT

• Graph paƩerns
Basic, group, opƟonal, alternaƟve, minus
Variable assignments
Filters

• SoluƟon modifiers
DISTINCT, REDUCED
GROUP BY, HAVING
ORDER BY, LIMIT, OFFSET

